

N=9_Algebraic_proof

July 20, 2021

MULTIPLE EXTREMAL DISC-PACKINGS IN COMPACT HYPERBOLIC SURFACES

** CASE N = 9: ALGEBRAIC PROOF**

An NEC-group uniformizing a non-orientable surface of genus p=3 that contains at least two extremal 2-packings Summary of the particular data for this file. It all comes from the numerical experiments:

- N=9 in 1)
 - A particular set of side-pairing transformations in 6)
 - The two bananas: side-pairing 6, banana 0, sign +1 and side-pairing 8, banana 0, sign +1 in 8)
 - The point $z_0 0.714+0.451i$ is the second intersection point in 9) (indexed 1, since sagemath first index is 0)
 - The point w in 11) is not the center of the disc. The exponent j equals 3.
 - A particular set of products of side-pairings in 14)
-

Recall that $u = e^R$ where $\cosh R = \frac{1}{2 \sin \frac{\pi}{N}}$.

Therefore $u + u^{-1} = \frac{u^2 + 1}{u} = \frac{1}{\sin \frac{\pi}{N}}$, thus $\sin \frac{\pi}{N} u^2 - u + \sin \frac{\pi}{N} = 0$ and we have

$$u = \frac{1 + \sqrt{1 - 4 \sin^2 \frac{\pi}{N}}}{2 \sin \frac{\pi}{N}}.$$

```
[1]: #Approximate value of u
N=9
print (e^(arccosh(1/(2*sin(pi/N))))).n(), (( 1+sqrt( 1-4*(sin(pi/N))^2 ) )/
→(2*sin(pi/N))).n()
```

2.52827832631621 2.52827832631621

1) We compute the minimal polynomial of u , assigned to the variable pol .

[2] :

```
N=9  
pol=(( 1+sqrt( 1-4*(sin(pi/N))^2 ) )/(2*sin(pi/N))).minpoly()  
show(pol)
```

```
x^12 - 6*x^10 - x^8 - 28/3*x^6 - x^4 - 6*x^2 + 1
```

[3] :

```
#Approximate value of the roots of pol: notice that u is the largest real root  
show(pol.roots(ring=QQbar))
```

```
[(-2.528278326316213?, 1),  
 (-0.3955260738468752?, 1),  
 (0.3955260738468752?, 1),  
 (2.528278326316213?, 1),  
 (-0.7778619134302061? - 0.6284352342403300?*I, 1),  
 (-0.7778619134302061? + 0.6284352342403300?*I, 1),  
 (-0.5077133059428725? - 0.8615260872246175?*I, 1),  
 (-0.5077133059428725? + 0.8615260872246175?*I, 1),  
 (0.5077133059428725? - 0.8615260872246175?*I, 1),  
 (0.5077133059428725? + 0.8615260872246175?*I, 1),  
 (0.7778619134302061? - 0.6284352342403300?*I, 1),  
 (0.7778619134302061? + 0.6284352342403300?*I, 1)]
```

2) We denote the field $\mathbb{Q}(u)$ as $K1$.

[4] :

```
K1.<u> = NumberField(pol,embedding=2)  
CC(u)
```

#The instruction CC gives the approximate value of an element of the field K1.
#The given embedding into the reals is chosen so that CC(u) gives the right value

[4] : 2.52827832631621

3) Since $u = e^R$, we see that $\sin \frac{\pi}{N}$ is the field element given by $\frac{u}{u^2 + 1}$. We call it sn .

[5] :

```
print(CC(u/(u^2+1)),sin(pi/N).n()) #A numerical verification  
sn=u/(u^2+1)  
show(sn) #Displays sn as an element of K1
```

```
(0.342020143325227, 0.342020143325669)
```

```
-3/64*u^11 + 21/64*u^9 - 9/32*u^7 + 23/32*u^5 - 43/64*u^3 + 61/64*u
```

4) Accordingly, $\cos^2 \frac{\pi}{N} = 1 - \sin^2 \frac{\pi}{N} = 1 - \frac{u^2}{(u^2+1)^2}$ is the field element given by $\frac{u^4+u^2+1}{(u^2+1)^2}$.

We call $v = \cos \frac{\pi}{N}$. It is an element of $\mathbb{Q}(u)$ if and only if $u^4 + u^2 + 1$ is a square in K1. We also denote it as cs as a nickname.

We denote $L = \mathbb{Q}(u, v)$. This field agrees with K1 when $v \in K1$, of course.

[6]: `print(CC(sqrt(u^4+u^2+1)/(u^2+1)),(cos(pi/N)).n()) #Numerical verification`

```
if (u^4+u^2+1).is_square():
    v=sqrt(u^4+u^2+1)/(u^2+1) # cos(pi/N)
    print(CC(v))
    cs=v
    show(v) #Displays v as an element of K1
    L=K1
else:
    Ring.<t> = PolynomialRing(K1)
    L.<v> = K1.extension(t^2-(u^4+u^2+1)/((u^2+1)^2))
    cs=v

#The code below this line constructs the right embedding of L into the complex numbers
#that matches the numerical values we know that u and v must have.
#We call this embedding L_approx
#It will be very important later, in particular in points 8) and 11)
j=0
test=0
while test!=1:
    L_approx = L.embeddings(CC)[j]
    if abs(CC(sqrt(u^4+u^2+1)/(u^2+1))-L_approx(v))<10^-3 and
       abs(CC(u)-L_approx(u))<10^(-3): test+=1
    j+=1
print L_approx(u),L_approx(v) #This line must give the values we knew in advance
#(numerical verification)
```

(0.939692620785057, 0.939692620785908)

0.939692620785057

$-3/16*u^{10} + 33/32*u^8 + 3/4*u^6 + 31/16*u^4 + 11/16*u^2 + 25/32$

2.52827832631621 0.939692620785966

5) The matrices of a, b, c are

$$a = \begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}, \quad b = \begin{pmatrix} 0 & u^{-1} \\ u & 0 \end{pmatrix}, \quad c = \begin{pmatrix} -\cos \frac{\pi}{N} & \sin \frac{\pi}{N} \\ \sin \frac{\pi}{N} & \cos \frac{\pi}{N} \end{pmatrix}$$

```
[7]: a=matrix(2,[1,0,0,-1])
      b=matrix(2,[0,u^(-1),u,0])
      c=matrix(2,[-cs,sn,sn,cs])
      show(a)
      show(b)
      show(c)
```

```

[ 1   0]
[ 0 -1]

[                                     0 -u^11 + 6*u^9 + u^7 + 28/3*u^5 + u^3 + u
 $\hookrightarrow$  6*u]
[                               u
 $\hookrightarrow$  0]

[      3/16*u^10 - 33/32*u^8 - 3/4*u^6 - 31/16*u^4 - 11/16*u^2 - 25/32 -3/64*u^11 + u
 $\hookrightarrow$  21/64*u^9 - 9/32*u^7 + 23/32*u^5 - 43/64*u^3 + 61/64*u]
[ -3/64*u^11 + 21/64*u^9 - 9/32*u^7 + 23/32*u^5 - 43/64*u^3 + 61/64*u      -3/16*u^10
 $\hookrightarrow$  + 33/32*u^8 + 3/4*u^6 + 31/16*u^4 + 11/16*u^2 + 25/32]

```

6) We introduce in the list s the side-pairings for this surface group: our conjecture, driven by numerical experiments, is that the surface group thus generated will contain more than one extremal packing.

We use the notation $G_i = (ca)^i(abcb)(ca)^{-i}$.

```
[8]: def G(i):
    return (c*a)^i*(a*b*c*b)*(c*a)^(-i)

s=['zero'] #for s[j] to denote the j-th side-pairing we add an "empty" item in
           ↴position 0
s.append((c*a)*b) #s1
s.append((c*a)^2*(a*b)*(c*a)) #s2
s.append(G(3)*(c*a)^5) #s3
s.append(G(3)^3*(c*a)^3*a*(c*a)^3) #s4
s.append(G(3)^5*(c*a)^7) #s5
s.append(G(3)^(-2)*(c*a)^3*a*(c*a)^5) #s6
s.append(G(3)^(-3)*(c*a)^3*b*(c*a)^(-3)*G(3)) #s7
s.append(G(3)^4*(c*a)^3*b*(c*a)^(-3)*G(3)^(-2)) #s8

show(s[1:]) # How do the matrices look like? Their coefficients lie in L, but
           ↴are they too horrible?
```

```

[
[ -3/64*u^10 + 21/64*u^8 - 9/32*u^6 + 23/32*u^4 - 43/64*u^2 - 3/64 25/32*u^11 - 9/
-2*u^9 - 29/16*u^7 - 193/24*u^5 - 87/32*u^3 - 43/8*u] [ -9/64*u^10 + 57/
-64*u^8 - 3/32*u^6 + 27/32*u^4 + 47/64*u^2 + 49/64 35/64*u^11 - 207/64*u^9 - 25/
-32*u^7 - 517/96*u^5 - 81/64*u^3 - 259/64*u] [ -9/64*u^10 + 57/64*u^8 - 3/
-32*u^6 + 27/32*u^4 + 47/64*u^2 + 49/64 35/64*u^11 - 207/64*u^9 - 25/32*u^7 - 517/96*u^5 - 81/64*u^3 - 259/64*u] [ -3/8*u^10 + 9/4*u^8
+ 3/8*u^6 + 7/2*u^4 + 11/4 15/32*u^11 - 45/16*u^9 - 3/8*u^7 - 79/16*u^5 - 27/32*u^3 - 7/2*u] [ -9/32*u^10 + 51/32*u^8 + 3/4*u^6 + 27/8*u^4 +
-25/32*u^2 + 89/32 7/64*u^11 - 33/64*u^9 - 29/32*u^7 - 143/96*u^5 - 61/
-64*u^3 - 37/64*u] [ 15/64*u^10 - 93/64*u^8 + 3/32*u^6 - 73/32*u^4 - 53/
-64*u^2 - 81/64 1/32*u^11 - 3/16*u^9 - 1/8*u^7 + 13/48*u^5 + 11/32*u^3 +
-1/2*u] [ 15/64*u^10 - 93/64*u^8 + 3/32*u^6 - 73/32*u^4 - 53/64*u^2 - 81/64
- 1/32*u^11 - 3/16*u^9 - 1/8*u^7 + 13/48*u^5 + 11/32*u^3 + 1/2*u] [ 45/
-64*u^10 - 273/64*u^8 - 15/32*u^6 - 201/32*u^4 - 47/64*u^2 - 253/64 11/8*u^11
- 261/32*u^9 - 31/16*u^7 - 77/6*u^5 - 39/16*u^3 - 267/32*u]
[ 3/16*u^11 - 33/32*u^9 - 3/4*u^7 - 31/16*u^5 - 11/16*u^3 - 25/32*u -3/64*u^10 +
-21/64*u^8 - 9/32*u^6 + 23/32*u^4 - 43/64*u^2 + 61/64], [ 3/64*u^11 - 15/
-64*u^9 - 9/32*u^7 - 23/32*u^5 - 49/64*u^3 - 35/64*u -15/64*u^10 + 99/64*u^8
- 21/32*u^6 + 73/32*u^4 - 39/64*u^2 + 107/64], [ 3/64*u^11 - 15/64*u^9 - 9/
-32*u^7 - 23/32*u^5 - 49/64*u^3 - 35/64*u -15/64*u^10 + 99/64*u^8 - 21/32*u^6
+ 73/32*u^4 - 39/64*u^2 + 107/64], [3/2*u^11 - 279/32*u^9 - 51/16*u^7 - 115/
-8*u^5 - 57/16*u^3 - 309/32*u -3/8*u^10 + 9/4*u^8 + 3/
-8*u^6 + 7/2*u^4 + 7/4], [129/64*u^11 - 759/64*u^9 - 111/32*u^7 - 599/32*u^5 -
-275/64*u^3 - 811/64*u -3/32*u^10 + 15/32*u^8 + 3/4*u^6 + 1/2*u^4 +
-19/32*u^2 + 25/32], [-3/2*u^11 + 279/32*u^9 + 51/16*u^7 + 115/8*u^5 + 57/16*u^3
+ 293/32*u -9/64*u^10 + 63/64*u^8 - 21/32*u^6 + 27/32*u^4 - 45/64*u^2 + 11/
-64], [-3/2*u^11 + 279/32*u^9 + 51/16*u^7 + 115/8*u^5 + 57/16*u^3 + 293/32*u
-9/64*u^10 + 63/64*u^8 - 21/32*u^6 + 27/32*u^4 - 45/64*u^2 + 11/64], [ -75/
-32*u^11 + 219/16*u^9 + 9/2*u^7 + 365/16*u^5 + 187/32*u^3 + 29/2*u -51/64*u^10 +
-303/64*u^8 + 33/32*u^6 + 247/32*u^4 + 145/64*u^2 + 323/64]
]

```

7) We compute $\cosh^2 \frac{d_n}{2}$ for the distances d_0, d_1, d_2 associated to a tiling of hyperbolic space with regular N-gons of angle $2\pi/3$. We store the three values in the list ch.

First, as $d_0 = 2R$, then $\cosh^2 \frac{d_0}{2} = \cosh R = \frac{(u^2+1)^2}{4u^2}$.

Second, $\cosh d_1 = \cosh^2(d_0) - \sinh^2(d_0) \cos(4\pi/N) = (1 - \cos(4\pi/N)) \cosh^2(d_0) + \cos(4\pi/N)$.

Now, since $\cosh d = 2 \cosh^2 \frac{d}{2} - 1$, we have

$$\cosh^2 \left(\frac{d_1}{2} \right) = \frac{1 + \cosh d_1}{2} = \frac{1 + (1 - \cos(4\pi/N)) \cosh^2(d_0) + \cos(4\pi/N)}{2}$$

that is

$$\cosh^2\left(\frac{d_1}{2}\right) = \frac{1 + \cosh d_1}{2} = \frac{1 + (1 - \cos(4\pi/N))(2 \cosh^2 \frac{d_0}{2} - 1)^2 + \cos(4\pi/N)}{2}.$$

Accordingly, we have $\cosh d_2 = \cosh^2(d_0) - \sinh^2(d_0) \cos(6\pi/N) = (1 - \cos(6\pi/N)) \cosh^2(d_0) + \cos(6\pi/N)$ and therefore

$$\cosh^2\left(\frac{d_2}{2}\right) = \frac{1 + \cosh d_2}{2} = \frac{1 + (1 - \cos(6\pi/N))(2 \cosh^2 \frac{d_0}{2} - 1)^2 + \cos(6\pi/N)}{2}.$$

NOTE: Recall that $\cos \frac{k\pi}{N}$ is the value at $x = \cos \frac{\pi}{N}$ of the k -th Chebyshev polynomial.

[9]:

```
ch=[(u^2+1)^2/(4*u^2)]
ch.append((1+(1-chebyshev_T(4,cs))*(2*ch[0]-1)^2+chebyshev_T(4,cs))/(2))
ch.append((1+(1-chebyshev_T(6,cs))*(2*ch[0]-1)^2+chebyshev_T(6,cs))/(2))
```

show(ch) # How do these values look like? They lie in L, but are they too \rightarrow horrible?

```
[-1/4*u^10 + 3/2*u^8 + 1/4*u^6 + 7/3*u^4 + 1/2*u^2 + 2,
 -5/8*u^10 + 57/16*u^8 + 7/4*u^6 + 149/24*u^4 + 15/8*u^2 + 73/16,
 -9/8*u^10 + 105/16*u^8 + 9/4*u^6 + 87/8*u^4 + 23/8*u^2 + 121/16]
```

8) We compute the equations of the two relevant bananas, and the coordinates of the intersection. We need to recall which bananas are the promising ones in each case (meaning for which side-pairing and which distance d_j) from the information we obtained numerically.

The two variables $sign$ can be (independently) 1 or -1 : we guess the right choice after computing (via CC) the numerical value of the intersection, that we expect to match what we obtained in our numerical experiments.

[10]:

```
def H(k,s): #A function that occurs in the equation of the banana
    if s.det() == -1:
        return 4*ch[k]
    elif s.det() == 1:
        return 4*(ch[k] - 1)

def banana(sign, s, k): #The equation itself: choosing sign=+1 or -1 gives each
    → of the two arcs
    polring=PolynomialRing(L, 'x,y')
    x, y=polring.0, polring.1
    A,B,C,D=s[0][0], s[0][1], s[1][0], s[1][1]
    return
    → sign*((B-C)*(1+x^2+y^2)+2*x*(A-D)-2*y*(B+C))-(1-x^2-y^2)*sqrt(H(k,s)-(A+D)^2+4*det(s))
```

```

##### Defining the equations
var('x,y')
sign1,sign2=1,1 #Choose the right couple of signs to point to the right
→component of each banana!
eq1=banana(sign1, s[6], 0) #Insert here the right number of sidepairing s[j]
→for this case!!!!!
eq2=banana(sign2, s[8], 0) #The third number is either 0, 1, or 2 and indicates
→which banana we want.
#Choose the right ones for this case!!!
show(eq1)
show(eq2)
#####

##### A bit of algebraic geometry here: J is the variety determined by both
→polynomials. We check first that
##### it has dimension 0: it consists of finitely many points. Finally we compute
→those points.
polring=PolynomialRing(L, 'x,y')
J = polring.ideal(eq1, eq2)
print J.dimension()
J.variety(L)

```

$$(33/16*u^{11} - 189/16*u^9 - 45/8*u^7 - 151/8*u^5 - 87/16*u^3 - 197/16*u)*x^2 + (33/16*u^{11} - 189/16*u^9 - 45/8*u^7 - 151/8*u^5 - 87/16*u^3 - 197/16*u)*y^2 + (3/4*u^{10} - 39/8*u^8 + 3/2*u^6 - 25/4*u^4 - 1/4*u^2 - 23/8)*x + (47/16*u^{11} - 273/16*u^9 - 49/8*u^7 - 703/24*u^5 - 125/16*u^3 - 309/16*u)*y + (u^{11} - 6*u^9 - u^7 - 28/3*u^5 - u^3 - 5*u)$$

$$(17/4*u^{11} - 99/4*u^9 - 35/4*u^7 - 485/12*u^5 - 21/2*u^3 - 53/2*u)*x^2 + (17/4*u^{11} - 99/4*u^9 - 35/4*u^7 - 485/12*u^5 - 21/2*u^3 - 53/2*u)*y^2 + (3*u^{10} - 18*u^8 - 3*u^6 - 28*u^4 - 6*u^2 - 18)*x + (31/16*u^{11} - 177/16*u^9 - 41/8*u^7 - 479/24*u^5 - 109/16*u^3 - 197/16*u)*y + (51/16*u^{11} - 303/16*u^9 - 33/8*u^7 - 247/8*u^5 - 97/16*u^3 - 307/16*u)$$

0

[10]: [{}y: 3/64*u^{10} - 21/64*u^8 + 9/32*u^6 - 23/32*u^4 + 43/64*u^2 - 29/64, x: 15/64*u^{11} - 75/64*u^9 - 57/32*u^7 - 55/32*u^5 - 77/64*u^3 - 87/64*u}, {}y: 129/2368*u^{10} - 675/2368*u^8 - 369/1184*u^6 - 557/1184*u^4 - 1503/2368*u^2 - 803/2368, x: 219/2368*u^{11} - 1047/2368*u^9 - 873/1184*u^7 - 1343/1184*u^5 - 2969/2368*u^3 - 3739/2368*u}]

9) We apply L_approx to the points we have obtained in order to determine which is the one that corresponds to the value we expected from our numerical experiments. If none of them

seems to be correct we need to go back to 8) and change the signs.

```
[11]: #WARNING: The numerical experiments gave z~0.714+0.451i in this surface
for point in J.variety(L):
    print L_approx(point[x]),L_approx(point[y])
```

```
0.631717234748492 0.364722115533852
0.714302207008913 0.451272035332311
```

10) We call zx and zy the real and imaginary part of the right intersection of bananas. The point $z = zx + izy$ is, still conjecturally, the center of a disc of a second (hidden) extremal packing.

```
[12]: zx, zy=J.variety(L)[1][x], J.variety(L)[1][y]
#Choose J.variety(L)[S] for the right value of S checking the output of 9)

show(zx) #Again having a look. Are they very large expressions in u?
show(zy)
```

```
219/2368*u^11 - 1047/2368*u^9 - 873/1184*u^7 - 1343/1184*u^5 - 2969/2368*u^3 - 3739/2368*u
```

```
129/2368*u^10 - 675/2368*u^8 - 369/1184*u^6 - 557/1184*u^4 - 1503/2368*u^2 - 803/
2368
```

11) Our numerical experiments suggest that the following center of the initial k-packing will have an important role:

$$w = \frac{u^2 - 1}{u^2 + 1} \left(\sin \frac{2\pi j}{N} - i \cos \frac{2\pi j}{N} \right)$$

for certain integer j (we need to get the right one in each case from the experiments).

(there is also an special case possible in which $w = 0$, the center of the unit disc).

We compute $\cos \frac{2\pi j}{N}$ as the value at $x = \cos \frac{\pi}{N}$ of the $2j$ -th Chebyshev polynomial. Then, we compute $\sin \frac{2\pi j}{N} = \sqrt{1 - \cos^2 \frac{2\pi j}{N}}$.

Important.- By the last expression we mean *the positive root*, but we don't know which of the two possible roots will Sage associate to the symbols $\sqrt{}$ and $-\sqrt{}$ applied to a given element of the field L . We need again to make use of L_approx to decide which one is the positive one (the warning is motivated for the fact that it may happen that the positive one is the one sage calls $-\sqrt{}$).

```
[13]:
```

```

#w_is_center='yes' #leave uncommented the right line. w_is_center='yes' si w is
→the center of the unit disc
w_is_center='no'

if w_is_center=='yes':
    wx=0
    wy=0
elif w_is_center=='no':
    j=3 #Enter here the right value for j in this case
    wy=-(u^2-1)*chebyshev_T(2*j,cs)/(u^2+1)
    wx=(u^2-1)*(sqrt(1-chebyshev_T(2*j,cs)^2))/(u^2+1)
    if L_approx(wx)<0: wx=-wx
    show(wx) #Looking at the expressions as polynomials in u or u and v
    show(wy)
    print L_approx(wx), L_approx(wy) #Numerical values

```

$15/64u^{11} - 75/64u^9 - 57/32u^7 - 55/32u^5 - 77/64u^3 - 87/64u$
 $3/64u^{10} - 21/64u^8 + 9/32u^6 - 23/32u^4 + 43/64u^2 - 29/64$
 $0.631717234748492 \quad 0.364722115533852$

12) We define the function $g(z) = \frac{z-w}{\bar{w}z-1} = \frac{(z-w)(w\bar{z}-1)}{|w|^2|z|^2-2\operatorname{re}(w\bar{z})+1} = \frac{x+iy-wx-iy}{(wx-iy)(x+iy)-1}$, but rather as a function $\mathbb{R}^2 \rightarrow \mathbb{R}^2$. This is for technical reasons (L has no chosen embedding into the complex numbers, so we couldn't perform in principle products as $u \cdot i$ that we will later need).

We need to compute the point $p = \frac{g(zx+izy)}{1+\sqrt{1-|zx+izy|^2}}$, but the denominator is never in L because $1-|zx+izy|^2$ is never a square. So we extend the field L to $K2 = L(w)$, where $w = \sqrt{1-|zx+izy|^2}$.

Finally $m = g(p)$ may be, according to our numerical experiments, the fixed point of an order two elliptic element inducing an automorphism of the surface.

```

[14]: def g(x,y):
    denom=(x*wx+y*wy-1)^2+(y*wx-x*wy)^2
    num1=(x^2+y^2)*wx-x*(wx^2-wy^2)-2*y*wx*wy-x+wx
    num2=(x^2+y^2)*wy-2*x*wx*wy+y*(wx^2-wy^2)-y+wy
    return [num1/denom, num2/denom]

show((1-(g(zx,zy)[0])^2-(g(zx,zy)[1])^2)) #Inspecting how does w^2 look like

if (1-(g(zx,zy)[0])^2-(g(zx,zy)[1])^2).is_square():
    print 'Warning: apparently w IS in the field L' #Apparently this never
→happens in our files

```

```

#If it happens the last lines of this cell would produce an error, since u
→K2_approx
#wouldn't be defined
else:
    Ring.<y> = PolynomialRing(L)
    K2.<omega> = L.extension(y^2-(1-(g(zx,zy)[0])^2-(g(zx,zy)[1])^2))
    px=g(zx,zy)[0]/(1+omega)
    py=g(zx,zy)[1]/(1+omega)
    mx=g(px,py)[0]
    my=g(px,py)[1]
    show(mx) #These are the coordinates of the fixed point of tau as elements of u
→the field L.
    show(my)
    #Again choosing a precise embedding of K2 into the complex numbers
    #This is not really needed, but allows the computation of numerical values
→for mx, my
    j=0
    test=0
    while test!=1:
        K2_approx = K2.embeddings(CC)[j]
        if abs(CC(sqrt(u^4+u^2+1)/(u^2+1))-K2_approx(v))<10^-3 and
→abs(CC(u)-K2_approx(u))<10^(-3) and K2_approx(omega)>0: test+=1
        j+=1
        print K2_approx(u),K2_approx(v), K2_approx(omega) #This line must give the
→values we knew in advance (numerical verification)

print K2_approx(mx), K2_approx(my)

```

```

9/4*u^10 - 33/4*u^8 - 135/4*u^6 - 63/2*u^4 - 22*u^2 + 21/4
(-241/1744*u^11 + 1599/1744*u^9 - 755/1744*u^7 + 7459/5232*u^5 - 61/218*u^3 + 235/
←436*u)*omega + 321/872*u^11 - 1815/872*u^9 - 525/436*u^7 - 349/109*u^5 - 805/
←872*u^3 - 1731/872*u
(3/3488*u^10 - 27/1744*u^8 + 9/218*u^6 + 193/1744*u^4 + 621/3488*u^2 + 85/
←872)*omega - 267/6976*u^10 + 1863/6976*u^8 - 195/3488*u^6 - 1481/3488*u^4 - 4911/6976*u^2 - 2901/6976
2.52827832626770 0.939692620816747 0.950488633190445
0.676705896506585 0.414595401193581

```

13) We compute the coordinates of $\bar{\tau}$

```
[15]: tau00=(-2*mx)/(1-mx^2-my^2)
tau01=(2*my+1+mx^2+my^2)/(1-mx^2-my^2)
tau10=(2*my-1-mx^2-my^2)/(1-mx^2-my^2)
```

```

tau11=(2*mx)/(1-mx^2-my^2)
tau=matrix(2,[tau00,tau01,tau10,tau11])
show(tau) # Once more we are curious. The entries lie in K2, but: how do they look like?

```

```

[(-1453/1088*u^11 + 8385/1088*u^9 + 1727/544*u^7 + 20819/1632*u^5 + 4207/1088*u^3 +
  9565/1088*u)*omega
   (-93/272*u^10 + 1113/544*u^8 + 105/
  -272*u^6 + 55/17*u^4 + 53/68*u^2 + 1127/544)*omega]
[ (279/272*u^10 - 3291/544*u^8 - 447/272*u^6 - 39/4*u^4 - 79/
  -34*u^2 - 3469/544)*omega (1453/1088*u^11 - 8385/1088*u^9 - 1727/544*u^7 -
  20819/1632*u^5 - 4207/1088*u^3 - 9565/1088*u)*omega]

```

14) We create a list `conj` containing what our numerical experiments suggested that should be c_j , the conjugate by τ of the side-pairing s_j , for each j . We finally check that $\tau s_j \tau^{-1} c_j^{-1}$ is trivial.

```
[16]: conj=['Zero'] #Create this list with the data from the numerical experiments!!!!
conj.append(s[5]*s[1]^(-1)*s[8]^(-2))
conj.append(s[5]*s[8]^(-2))
conj.append(s[7]^2*s[8]^(-2))
conj.append(s[7]*s[8]^(-2))
conj.append(s[3]*s[5]^(-1))
conj.append(s[8]^(-1))
conj.append(s[8]^(-1))
conj.append(s[6]^(-1))

for j in [1..len(s)-1]:
    show(tau*s[j]*tau^(-1)*conj[j]^(-1))
```

```

[-1  0]
[ 0 -1]
```

```

[-1  0]
[ 0 -1]
```

```

[-1  0]
[ 0 -1]
```

```

[1  0]
[0  1]
```

```

[-1  0]
[ 0 -1]
```

```

[1  0]
[0  1]
```

```

[1  0]
[0  1]
```

$[1 \ 0]$
 $[0 \ 1]$

[]: